

Lab 4 Report


EE 4111

Cole NIELSEN

Spring 2016

Spice File

Below is the SPICE netlist file and the schematic of the OP-AMP design used in this lab for the noise simulations. This is the same compensated OP-AMP designed in the previous lab sessions. Noise simulation was augmented in this file by adding a .NOISE SPICE directive to instantiate a noise simulation on parsing the file through HSPICE.



```
TWO STAGE OPAMP W/ NOISE SIMULATION
*****SIMULATION PARAMETERS*****
.OPTIONS LIST NODE POST
.AC DEC 100 0.1 1E6
.NOISE V(nvo) Vi2 100

.parameter Vid = 0
.parameter Vic = 0
.parameter Vios = -213.913u ***input offset
.parameter length = 0.25u

*****SUPPLIES*****
Vdd nvdd 0 DC 2V
Vss 0 nvss DC 2V
Vi1 ni1 0 DC Vios
Vi2 ni2 0 DC 0 AC 1 *PULSE(0 1 25n 10p 10p 50n 1u)
*Ei1 ni1 0 VOL = 'Vios + Vic + Vid'
*Ei2 ni2 0 VOL = 'Vic - Vid'

*****CURRENT REF*****
Rb nvt nvss 26000

M8 nbias nbias nvdd nvdd CMOSF W = 30u L = length
M9 n3 nbias nvdd nvdd CMOSF W = 16u L = length
```

```

M10 nbias n3 nvt nvss CMOSN W = 10u L = length
M11 n3 nvt nvss nvss CMOSN W = 10u L = length
*Mx D G S B CMOSx W = width L = length

```

```

*****DIFF AMP*****

```

```

M1 n1 ni1 n2 nvdd CMOSP W = 10u L = length
M2 nvdo ni2 n2 nvdd CMOSP W = 10u L = length
M3 n1 n1 nvss nvss CMOSN W = 4u L = length
M4 nvdo n1 nvss nvss CMOSN W = 4u L = length
M5 n2 nbias nvdd nvdd CMOSP W = 16u L = length

```

```

*****STAGE 2*****

```

```

M6 nvo nvdo nvss nvss CMOSN W = 2.3u L = length
M7 nvo nbias nvdd nvdd CMOSP W = 6u L = length
*C1 nvo 0 10p

```

```

*****COMPENSATION*****

```

```

Cc nvdo nx 0.24pF
M12 nvdd nvo nx nvss CMOSN W = 100u L = length
Ibias nx nvss 10u

```

```

.MODEL CMOSN NMOS (

```

+LEVEL = 49	acm = 3	hdif = 0.35e-6
+VERSION = 3.1	TNOM = 27	TOX = 5.7E-9
+XJ = 1E-7	NCH = 2.3549E17	VTH0 = 0.4365497
+K1 = 0.3915623	K2 = 0.0175145	K3 = 1E-3
+K3B = 2.6588343	W0 = 1E-7	NLX = 1.111465E-7
+DVTOW = 0	DVT1W = 0	DVT2W = 0
+DVT0 = -0.0408321	DVT1 = 0.0746768	DVT2 = 0.307109
+U0 = 407.1177485	UA = 9.442714E-11	UB = 1.092986E-18
+UC = 1.63196E-11	VSAT = 1.365087E5	A0 = 1.3189329
+AGS = 0.2711719	B0 = 3.291713E-8	B1 = -1E-7
+KETA = 4.645753E-3	A1 = 0	A2 = 1
+RDSW = 439.9558234	PRWG = 0.0345487	PRWB = -0.0441065
+WR = 1	WINT = 1.645705E-9	LINT = 1.116516E-9
+XL = 3E-8	XW = 0	DWG = -1.494138E-9
+DWB = 1.459097E-8	VOFF = -0.1026054	NFACTOR = 0.1344887
+CIT = 0	CDSC = 1.527511E-3	CDSCD = 0
+CDSCB = 0	ETA0 = 1.930311E-3	ETAB = 2.946158E-4
+DSUB = 0.0214865	PCLM = 1.3387947	PDIBLC1 = 0.480652
+PDIBLC2 = 9.034986E-3	PDIBLCB = -1E-3	DROUT = 0.5593223
+PSCBE1 = 9.843289E9	PSCBE2 = 2.10878E-9	PVAG = 1.0033136
+DELTA = 0.01	MOBMOD = 1	PRT = 0
+UTE = -1.5	KT1 = -0.11	KT1L = 0
+KT2 = 0.022	UA1 = 4.31E-9	UB1 = -7.61E-18
+UC1 = -5.6E-11	AT = 3.3E4	WL = 0
+WLN = 1	WW = -1.22182E-16	WWN = 1.2127

```

+WWL      = 0          LL      = 0          LLN      = 1
+LW        = 0          LWN     = 1          LWL      = 0
+CAPMOD    = 2          XPART   = 0.4        CGDO      = 3.11E-10
+CGSO      = 3.11E-10   CGB0    = 1E-11     CJ        = 1.758521E-3
+PB         = 0.99      MJ       = 0.457547   CJSW      = 4.085057E-10
+PBSW      = 0.8507757  MJSW   = 0.3374073  PVTHO     = 7.147521E-5
+PRDSW     = -67.2161633 PK2     = -1.344599E-3 WKETA     = 3.035972E-3
+LKETA     = -9.0406E-3 LAGS    = -0.3012   NLEV = 2 KF = 1E-26)
*
.MODEL CMOSP PMOS (
+LEVEL     = 49          acm      = 3          hdif      = 0.35e-6
+VERSION   = 3.1        TNOM     = 27         TOX       = 5.7E-9
+XJ        = 1E-7       NCH      = 4.1589E17    VTHO      = -0.6586391
+K1        = 0.5199897  K2      = 0.0357513    K3        = 0
+K3B       = 15.5613889 W0       = 1E-6        NLX       = 1E-9
+DVTOW     = 0          DVT1W   = 0          DVT2W     = 0
+DVT0      = 2.6100181 DVT1    = 0.4363142   DVT2      = -0.042436
+U0        = 196.024903 UA       = 2.767112E-9 UB        = 1.90709E-18
+UC        = 6.166867E-11 VSAT    = 1.975064E5   A0        = 0.2398712
+AGS       = 0.0943234  B0      = 3.21184E-6   B1        = 5E-6
+KETA      = 0.0312217  A1      = 0          A2        = 1
+RDSW      = 997.072701 PRWG     = -0.1916111  PRWB      = -0.495
+WR        = 1          WINT     = 2.527293E-9  LINT      = 1.254514E-8
+XL        = 3E-8       XW       = 0          DWG       = -3.253948E-8
+DWB       = 4.92072E-8 VOFF    = -0.15       NFACTOR   = 1.5460516
+CIT        = 0         CDSC     = 1.413317E-4  CDSCD     = 0
+CDSCB     = 0          ETA0     = 0.7241245   ETAB      = -0.240523
+DSUB      = 1.0813613 PCLM    = 2.0772083   PDIBLC1   = 4.31459E-4
+PDIBLC2   = 0.0252121 PDIBLCB = -9.960722E-4 DROUT     = 0.0432774
+PSCBE1    = 3.191047E10 PSCBE2 = 1.323218E-8  PVAG      = 0.0420525
+DELTA     = 0.01       MOBMOD  = 1          PRT       = 0
+UTE       = -1.5       KT1     = -0.11      KT1L      = 0
+KT2       = 0.022      UA1     = 4.31E-9    UB1       = -7.61E-18
+UC1       = -5.6E-11   AT      = 3.3E4      WL        = 0
+WLN       = 1          WW       = 0          WWN       = 1
+WWL       = 0          LL       = 0          LLN       = 1
+LW        = 0          LWN     = 1          LWL       = 0
+CAPMOD    = 2          XPART   = 0.4        CGDO      = 2.68E-10
+CGSO      = 2.68E-10   CGB0    = 1E-11     CJ        = 1.902493E-3
+PB         = 0.9810285 MJ       = 0.4644362   CJSW      = 3.142741E-10
+PBSW      = 0.9048624 MJSW   = 0.3304452   PVTHO     = 4.952976E-3
+PRDSW     = 29.8169373 PK2     = 3.383373E-3 WKETA     = -7.913501E-3
+LKETA     = -0.0208318 NLEV    = 2    KF = 1E-26)
*
.end


```

1 Basic OP-AMP Noise Analysis

The first objective of this lab was to analyze the output noise of the compensated OPAMP designed in the previous labs. This first set of simulations only factored in simple thermal (Johnson-Nyquist) noise, which is generated by all the resistors and MOSFETs of the circuit. Thermal noise is quantified by the following relation:

$$\overline{v_n^2} = 4k_bTR \quad (1)$$

Where $\overline{v_n^2}$ is the thermal noise variance, k_b is Boltzmann's constant, T is the absolute temperature given in Kelvin and R is the resistance of the noise generating medium. This relation tells us one important detail: thermal noise is dependent on the resistance of the devices for this simulation, as the temperature is assumed to be constant. Therefore, the resistors and MOSFETs of the circuit will exhibit thermal noise (the MOSFETs are resistive primarily due to their resistive channels). The noise analysis was performed in SPICE by adding a .NOISE statement that simulates noise for the output node nVo relative to an input source, for this simulation ni2 was used. The OP-AMP was simulated in an open-loop configuration with the other input ni1 connected to ground (with input offset correction included). The noise simulation was set to simulate for 0.1 Hz to 100 MHz in the .AC statement. The results of this simulation are presented below:




This plot gives total noise power spectral density of the output relative to the OPAMP input in units of $\frac{V}{\sqrt{Hz}}$, which is just the square root of the noise variance given in equation 1. It can be seen that the noise begins to drop off after approximately 10kHz, due to the pole location of the OP-AMP compensation. The shape of the response is effectively the transfer function for the input compensation as the noise given no poles should follow a uniform distribution, so the only changes for the PSD should be in accordance to the pole of the OP-AMP. Total RMS (root-mean-square) noise of the OP-AMP on the output was determined by integrating

the PSD over the simulated frequency range, where the interval of intergration is given by $[f_l, f_h]$:

$$V_{n,RMS} = \int_{f_l}^{f_h} \sqrt{v_n^2} df \quad (2)$$

Below is the plot showing the result of this integration. Total RMS Noise on the output was found to be $42.772\mu V$ on the ouput, if no other filtering is used and a bandwidth of 100 MHz is utilized.




2 Flicker Noise Simulation Modification


The second objective of this lab was to add flicker noise to the simulation for the OP-AMP. The provided mosfet models used in this lab do not factor in flicker ($1/f$) noise, so several parameters had to be added to include it in simulation. The basic model parameters in LEVEL 49 are NLEV, which is the noise equation selector, AF, the flicker noise exponent, and KF which is the flicker noise coefficient. NLEV was set to the default value of 2, therefore the standard noise simulation equations were used. AF and KF are parameters of the following flicker noise equation:

$$\overline{i_n^2} = KF \frac{I}{f_{AF}} \quad (3)$$

Where $\overline{i_n^2}$ is the flicker noise PSD for current (SPICE simulates in terms of current). The equivalent PSD in terms of voltage can be found by multiplying the current PSD by R^2 . Notice the flicker noise is directly proportional to DC current through an element, and inversely to the frequency. An important implication of this is the flicker noise decreases with frequency, eventually falling below the thermal noise. The intersection point of flicker noise is the transisition frequency. It should be noted because of this, flicker noise dominates at low frequencies. This effect is illustrated below (where f_a is the intersection frequency).




The values for KF was adjusted until the transition frequency between the two noise curves was approximately 10kHz, which is typical for MOSFETs. This value was found to be $KF = 1E-26$ for $AF = 1$. The PSD for output noise of the OPAMP including this flicker noise is shown below:




The total RMS noise on the output can be calculated by again integrating the PSD over the simulated frequency range, using the calculation tools built into COSMOSCOPE. Total RMS noise over the whole band analyzed was calculated to be approximately $67\mu V$, larger than the approximately $42\mu V$ seen without inclusion of flicker noise. Looking at the integrated PSD,

the two regions where flicker noise is dominant and then where thermal noise is dominant can be observed. The flicker noise region is the first flat line, and the thermal noise region is where the slope suddenly changes near 20kHz and then eventually tapers off due to the OP-AMP's compensation bringing the gain to essentially 0.



3 Input Referred Noise

The objective of this part of the lab was to find the equivalent referred input noise PSD for the OP-AMP design. Equivalent input referred noise is simply the noise that would be fed into the input of a noiseless OP-AMP in order to get the same output noise of that OP-AMP if it was noisy with no input signal. This is illustrated below, where $\overline{v_{n,i}^2}$ is the input referred noise variance and $\overline{v_n^2}$ is the measured output noise variance:



Since the output noise for the OP-AMP has been calculated from previous parts, the equivalent referred input noise can simply be calculated by dividing the output noise PSD by the OP-AMP gain. This gain was found to be approximately 750. Therefore, the PSD for equivalent input referred noise is the plot shown below, which was calculated in COSMOSCOPE by dividing the output noise PSD from part 2 by 750:

